


# 修平科技大學 電機工程系

DEPARTMENT OF ELECTRICAL ENGINEERING  
HSIUPING UNIVERSITY OF SCIENCE TECHNOLOGY

## 實務專題報告書

### 家庭智慧型保全系統規畫製作



指導老師：趙維和

專題製作學生：四技電五甲 周晉玄 BD96131

四技電五甲 林泉助 BD96001

中華民國 一 百 年 十 二 月 十 日

## 摘 要

本專題以8051為核心，透過掃描鍵盤以及LCD來設定，並利用溫度感測器、氣體感測器以及磁簧開關來感測家中有無異狀，如發現異常時，感測元件將通知8051觸發撥號IC，使撥號IC撥號給使用者，讓使用者能在第一時間將危害降到最低

# 目 錄

第一章 導論 .....	4
第二章 電路設計 .....	5
基本用法 .....	6
熱敏電阻 .....	8
LCD 簡介 .....	11
掃描鍵盤 .....	21
MQ-2 氣體感測器 .....	27
第三章 89C51程式規劃 .....	35
動作解說 .....	36
程式流程圖 .....	37
程式碼 .....	38
電路圖 .....	63
成品圖 .....	66
零件表 .....	67
附 錄 .....	69

# 第一章 導論

## 1-1 動機：

現今社會科技日新月異，機械產品電腦化已經是必然的趨勢，各式各樣的機械產品和新發明無非是要讓人們的生活更加方便舒適。就像現在人手一隻手機，這就大大地改善了傳統電話的缺點，手機不用在固定的地點經由電話線傳送，而直接可以經由訊號的傳遞，而與對方交談，所以現在的社會，已經可以說是進入到了無線的時代。

”機械自動化”的觀念已經普遍化，從工廠的機械自動化，再到辦公大樓的機械自動化，甚至現在的公寓大樓也採用晶片確認身份的方式，這些時代潮流的趨勢，使得”家電機械自動化”的觀念，也已經慢慢地深入人心。

## 1-2 目的：

此專題以8051單晶片微電腦系統為主元件架構，結合數種感測元件、LCD顯示器及撥號IC來完成家庭保全系統。

由溫度感測器、瓦斯感測器及磁簧開關檢測家裡有無異狀。當產生異狀感測器受到觸發，透過8051單晶片控制撥號IC來撥號通知家人及警消人員，這樣即使家人不在，亦可在家裡發生異狀時，馬上收到訊息，以達到家庭保全的功能。

## 第二章 電路設計

### 2-1-1 類比/數位轉換器 ADC0801~ADC0805的認識

雖然微電腦可以將各種數位資料做快速而精確的處理，但是人類在日常生活中所遇到的各種物理量(例如溫度、亮度、重量)都是類比的，因此欲令微電腦處理類比信號，必須先將類比信號轉換成數位信號才送進微電腦。

類比/數位轉換器(analog to digital converter)簡稱為A/D轉換器(A/D converter)。A/D轉換器的功能是將輸入之類比信號轉換成數位信號輸出。

目前最常用的A/D轉換器是ADC0801系列，編號從ADC0801到ADC0805，ADC0805，它們的接腳及特性皆相容，可以互相代換使用，各說明如下：

(1) 功能：8位元A/D轉換器

(2) 接腳圖：見圖2-1

(3) 特點：

※8位元 CMOS 連續漸進型 ADC。

※三態門鎖輸出。

※最大誤差 $\pm 1\text{LSB}$ 。

※轉換時間  $100\mu\text{S}@640\text{KHZ}$ 。

※內部含有時脈產生電路，頻率由外加 R、C 決定。




圖 2-1 ADC080X 接腳

## 2-1-2 基本用法

(1) 基本用法見圖2-2

(2) 動作情形：

a.若令CS及WR腳皆為”0”，則會令INTR接腳變為”1”，且使類比至數位  
電路完成準備工作。

b.經過100ns後，若CS與WR之中有任一腳以上變為”1”，則類比至數位  
轉換電路進入工作狀態，開始將類比輸入電壓 $V_x$ 轉換成8位元的數位  
資料。

c.當轉換好後，會將數位資料保存在栓鎖器，並令INTR腳輸出”0”表示  
已轉換完成，見圖2-3。




圖 2-2 ADC0804 基本用法




圖 2-3 時序圖

d. 若令 CS 及 RD 腳皆為"0", 則三態緩衝器導通, 將數位資料由 DB7~DB0 輸出, 見圖 2-4。




圖 2-4 時序圖

- e. 第 9 腳 ( $1/2 V_{REF}$ ) 這是一支很重要的輸入腳, 標示  $V_{REF}/2$ , 表示只要輸入  $V_{REF}$  的一半到 Pin9, 就能得到步階的大小  $= V_{REF} / 2^N = V_{REF} / 2^8 = V_{REF} / 256$ 。
- f. 第 10 和 8 腳 (D GND), (A GND) 分別是數位的接地及類比的接地。
- g. 數位輸出信號 DB 與類比輸入電壓  $V_x$  的關係為:  $DB = 51 V_x$ 。
- h. ADC0804 其內部已內建一組時脈產生電路, 只要在第 19 和 4 腳之間接一個電阻 R, 且於第 4 腳上再接一個電容 C 對地, 就能產生時脈供

ADC0804使用。而其頻率的大小約為 $f_{CLK} \doteq 1/(1.1RC)$

(3) 防雜訊的干擾：

a. 在第20和10腳之間接一個1uF 以上的電容。

b. 在第8和9腳之間接一個0.1uF 的電容。

## 2-2 熱敏電阻

量測溫度的高低，熱敏電阻是最常見且便宜的元件，英文名稱為 thermistor 簡寫為Th。熱敏電阻是一種電阻值對溫度極為敏感的半導體元件，分為兩種：正溫度係數及負溫度係數，而正溫度係數熱敏電阻(Positive Temperature Coefficient Thermistors；簡稱PTC熱敏電阻)為隨著溫度升高，電阻值增加的熱敏電阻，反之，負溫度係數熱敏電阻(Negative Temperature Coefficient Thermistors；簡稱NTC熱敏電阻)為隨著溫度升高，電阻值降低的熱敏電阻。

本專題我們將採用負溫度係數熱敏電阻NTC，特性如圖3-3所示。熱敏電阻的典型用法如圖3-4所示。本專題採用3-4 (a)圖， $V_x$ 的計算公式為分壓公式 $V_x = 5 * R_x / (R_x + R_{TH})$ ，實際實驗數值見表3-2




圖 3-3 熱敏電阻 NTC 特性圖




圖 3-4 熱敏電阻典型用法

溫度	電阻值	V <sub>x</sub> 電壓	A/D 值	範圍	四捨五入
18	1.45	2.040816	104.0816	105.0554	105
19	1.405	2.079002	106.0291	107.5017	108
20	1.34	2.136752	108.9744	110.0427	110
21	1.295	2.178649	111.1111	112.4752	112
22	1.24	2.232143	113.8393	114.8742	115
23	1.2	2.272727	115.9091	117.534	118
24	1.14	2.336449	119.1589	120.2937	120
25	1.1	2.380952	121.4286	122.8185	123
26	1.053	2.43546	124.2085	125.4425	125
27	1.013	2.483855	126.6766	128.0591	128
28	0.97	2.538071	129.4416	130.4425	130

29	0.94	2.57732	131.4433	132.6508	133
30	0.905	2.624672	133.8583	135.2937	135
31	0.865	2.680965	136.7292	138.0367	138
32	0.83	2.73224	139.3443	140.3093	140
33	0.805	2.770083	141.2742	142.4681	142
34	0.775	2.816901	143.662	145.1068	145
35	0.74	2.873563	146.5517	147.6199	148
36	0.715	2.915452	148.688	150.0117	150
37	0.685	2.967359	151.3353	152.4749	152
38	0.66	3.012048	153.6145	154.7889	155
39	0.635	3.058104	155.9633	156.9779	157
40	0.614	3.097893	157.9926		

表 3-2 熱敏電阻參考實驗數值表





表 3-3 溫度 A/D 值範圍表

## 2-2-1 LCD 簡介

液晶顯示器,因為工作所需的電流非常低,因此被廣泛使用於高級電器產品及儀表的顯示裝置上.目前市面上的 LCD,大都已經包裝成模組以方便控制,可分文字型及繪圖形 2 種.文字型 LCD有16X1,16X2,20X1,20X2,40X2數種.

### LCD 模組的結構

圖 2.5 LCD 模組結構



LCD 模組是由控制器,驅動器,顯示器三部份所組成,而目前市面上 LCD的模組驅動/控制器都是由日製的HD44780控制器晶片(或是相容產品),因此外部接腳的功能也都相同,這個控制器的內部共有 80 個位元組可供儲存由外部送進來的資料,因此使用此款控器的LCD模組最多一次可顯示 80 個字.

HD44780 中,儲存顯示資料的 80 個位元組,稱之為 Display Data

RAM(DDRAM),DDRAM 的位址從 00H~67H,共 80 個 Byte.

在 LCD 上 Show 出資料

DDRAM 位址與可供顯示位置的對應圖

	0	1	2	3	.....	12	13	14	15
第一行	80	81	82	83	.....	8C	8D	8E	8F
第二行	C0	C1	C2	C3		CC	CD	CE	CF

16 字 X 2 行 32 字

至於 CPU 如何將資料送給 LCD 顯示的方法,是將欲顯示的字元的ASCII碼寫到 LCD 內部的 DDRAM,LCD 就會將這個自在其對應的位置顯示出來.例如,若想在 LCD 的左上角處(位址 80h),顯示'B'.就將 ASCII碼-42H,送到 DDRAM 的 80H 位址.


例:

```

MOV     A,#80H           ;設定要顯示的位置為第一行第一字
        ACALL  W_INS      ;呼叫副程式，將指令傳至LCD
MOV     A,#"B"          ;顯示值為"B"
        ACALL  W_DATA     ;呼叫副程式，將資料傳至LCD
    
```

## 2-2-2 LCD 的接腳規格

雖然 LCD 的接腳因製造廠商的不同而排列方式有所不同,不過都是編號由 1 至 14,的 14 根接腳.



NO	Symbol	Function
1	VSS	主電源 0V
2	VDD	接地端 5V
3	V0	顯示明暗對比控制腳 <i>CONTRAST ADJ</i>
4	RS	暫存器選擇信號(Register) RS=0 為資料,RS=1 為指令暫存器
5	R/W	R/W=1 表示從 LCD 模組讀取資料 R/W=0 表示將資料寫入 LCD 模組
6	E	LCD 模組的致能端
7	DB 0	<i>Data Bit 0</i>
8	DB 1	<i>Data Bit 1</i>
9	DB 2	<i>Data Bit 2</i>
10	DB 3	<i>Data Bit 3</i>
11	DB 4	<i>Data Bit 4</i>
12	DB 5	<i>Data Bit 5</i>
13	DB 6	<i>Data Bit 6</i>
14	DB 7	<i>Data Bit 7</i> 但是也拿來傳送 BF

## 2-2-3 LCD內部的旗號與暫存器

### 1. 忙碌旗號 Busy Flag(BF)

BF 旗號是 LCD 用來告訴 CPU 他內部是否再忙碌的一個旗號,若 BF=1,表示 LCD 正在處理內部的工作,因此此時不可以寫資料給 LCD.當BF=0,表示 CPU 可以寫資料給 LCD.當接腳 RS=0 且 R/W=1 時,忙碌旗號的接腳會由 DB7 輸出.

### 2. 暫存器

LCD 模組內只有 2 個 8 位元暫存器,稱為指令暫存器(Instruction Register IR),和資料暫存器(Data Register,DR),他們都是 8 位元暫存器,由 RS 腳來選用.

接 腳		作 用
RS	R/W	
0	0	把指令碼或位址寫入指令暫存器 IR,並執行指令或設定一個要存入/取出資料的位址.
0	1	讀取忙碌旗標 BF 或位址計數器 AC 的內容. DB=7 忙碌旗標 BF 的內容. DB6~DB0=位址計數器 AC 的內容
1	0	把資料寫入資料暫存器 DR 內部會自動執行 DR->DD RAM 或 DR->CG RAM
1	1	從資料暫存器 DR 讀取資料 內部會自動執行 DR<-DD RAM 或 DR<-CG RAM

指令暫存器 IR 用來接收單晶片送來的命令,例如清除顯示,或功能設定等等...資料暫存器 DR 則用來接收單晶片要寫到 DDRAM(共 80Byte)或 CGRAM(共 64Byte)的資料緩衝區.當單晶片寫到 DR 暫存器之後.LCD 內部的控制電路會將資料自動寫到 DD RAM 或 CG RAM 中,而位址是由 LCD 內部的位址計數器(Address Counter AC)所指定.而單晶片要讀取資料時,需先將欲讀取的位址放入 IR 暫存器中,LCD 就會將其內容放入 DR 中,然後單晶片就可以去讀取 DR 的資料.

### 3. 位址計數器 AC

位址計數器是用來指定欲存取的 DD RAM和 CG RAM 的位址.位址設定


指令將位址寫入 IR 暫存器之後,LCD 內部控制電路會將 IR 暫存器的內容送至 AC.當資料存取之後,AC 便會自動加一(I/D=1)或減一(I/D=0).

#### 2-4- 字元產生器 Character Generat ROM (CG ROM)


LCD 內部有一個存放字型的 ROM,它裡面存著 192 個 5x7 點矩陣的字型.這些字型,由存放在 DD RAM 中的 ASCII 碼叫出來顯示.例如:'A'的字型碼為 01000001(即 41H)

#### 5.自創字型產生器

LCD 模組除了提供標準的字型 CG ROM 供人使用之外,另外還提供一塊 64 位元組的 CG RAM 空間供人從放 user 自己設計的字,一個字要 8個位元組(5x7 點矩陣),因此  $64/8=8$ ,做多可放 8 個新字元.要顯示自己的字元時,需先將 5X 點矩陣圖形放入 CG RAM 中.而要叫出自元時,其字型碼為 00H~07H.



↑ 圖2-4-4 LCD 模組寫入時序



↑ 圖2-4-5 LCD 模組讀出時序圖

## 2-2-4 LCD 模組控制命令

LCD 模組可以接受 CPU 送至 IR 暫存器的命令,並加以執行.其指令共有 11 道,除了 Clear Display 和 Cursor Home 為 1.64ms 之外,其餘都為 40ns.

### 1. 清除顯示 Clear Display

指令

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
0	0	0	0	0	0	0	0	0	1

動作:將 DDR 資料全部填入空白碼 20H,並將游標移到左上角原點位置,清除 DD RAM 的位址計數器 AC 為 00H,I/D 設為 1.

### 2. 游標歸位 Curuse Home

指令

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
0	0	0	0	0	0	0	0	1	*

動作:DD RAM的資料保持不變,僅將游標移至左上角原點處,及第一行的第一個字,DD RAM 的位址計數器設為 00H.

### 3. 輸入模式設定 Entry Mode

指令

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
0	0	0	0	0	0	0	0	I/D	S

動作:這個指令設定游標的移動方向,和顯示器是否要閃爍.

S	I/D	動作說明
0	0	顯示字元不動,游標左移,位址計數器 AC 減一
0	1	顯示字元不動,游標右移,位址計數器 AC 加一
1	1	顯示字元右移,游標不動,位址計數器 AC 不變
1	1	顯示字元左移,游標不動,位址計數器 AC 不變

#### 4. 顯示 ON/OFF 控制(Display ON/OFF)

指令

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
0	0	0	0	0	0	1	D	C	B

動作: D:顯示器控制位元,D=0 關 D=1,開

C:游標顯示控制位元,C=0 不顯示 C=1 顯示

B:游標閃爍 Blink 控制位元,B=0 不閃爍,B=1 閃爍

#### 5. 游標移位控制

指令

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
0	0	0	0	0	1	S/C	R/L	*	*

動作: 這個指令可以在不做任何讀寫資料的情況下,移動游標或整個顯示幕向右或向左移.

S/C	R/L	動作說明
0	0	游標左移,AC 值減
0	1	游標右移,AC 值加
1	1	整個顯示幕向左移(所有資料和游標)
1	1	整個顯示幕向右移(所有資料和游標)

#### 6. 功能設定 Function Set 指令

動作: DL:設定介面的位元寬度,DL=1 時為 8 位元,若 DL=0 時為 4 位元.

N:設定顯示器的行數,N=0 時,1 行顯示 N=1 時,2 行顯示.

F:字型 Font 設定.

#### 7. CG RAM 位址設定指令

動作:設定 CG RAM 的位址,由 A5~A0 之六個位元位址 00H~3FH 位址共 64 個位元組,當寫入本命令之後,接個輸入的資料將寫到 CG RAM 中.

### 8. DD RAM 位址設定指令

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
0	0	1	A6	A5	A4	A3	A2	A1	A0

動作:設定 DD RAM 的位址,由 A6~A0 之七個位元定址,接著寫入的資料將寫入 DD RAM 中.

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
0	1	BF	A6	A5	A4	A3	A2	A1	A0

### 9. 讀取忙碌旗標 BF 即位址計數器 AC 的內容指令

動作:若讀出的 BF=1,表示 LCD 模組正忙著內部的工作,因此,無法接受 CPU 的命令,直到 BF=0 才可以.在讀 BF 的同時,也會讀到位址計數器的值(DB0~DB6),這個讀出的位址,可能是 CG RAM 的位址或是 DD RAM 的位址,端看前面設定.

### 10. 寫資料到 CG RAM 或 DD RAM 中指令

動作:寫入 8 位元的資料(D0~D7)到 DD RAM或 CG RAM,至於是寫到那

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
1	0	D7	D6	D5	D4	D3	D2	D1	D0

一個,則是看前一次式設定 DD RAM或是 CG RAM 的位址.

### 11. 自 CG RAM 或 DD RAM 中讀取資料指令

動作:寫入 8 位元的資料(D0~D7)到 DD RAM或 CG RAM,至於是寫

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB3	DB1	DB0
1	1	D7	D6	D5	D4	D3	D2	D1	D0


到那一個,則是看前一次式設定 DD RAM 或是 CG RAM 的位址

圖 2-4-6 LCD 控制指令表

命 令 (Instruction)	指 令 編 碼										功 能 說 明
	RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0	
清除顯示器	0	0	0	0	0	0	0	0	0	1	將所有顯示的資料清除，且游標回到原點（位址 00H）
游標歸位	0	0	0	0	0	0	0	0	1	*	游標回到原點，但 DD RAM 的內容不變
進入模式設定	0	0	0	0	0	0	0	1	I/D	S	設定游標移動方向，或讀/寫資料後是否要移標位址是否閃爍 (B)
顯示器 ON/OFF 控制	0	0	0	0	0	0	1	D	C	B	控制所有顯示器 (D)，或游標 (C) 的 OF/OFF，和游標位址是否閃爍 (B)
顯示器或游標移動方式設定	0	0	0	0	0	1	S/C	R/L	*	*	在不需要改變 DD RAM 內容的情況下，移動游標和顯示器資料
功能設定	0	0	0	0	1	DL	N	F	*	*	設定界面匯流排寬度 (DL) 顯示器行數 (N)，和字形 (F)
CG RAM 位址設定	0	0	0	1	AC5	AC4	AC3	AC2	AC1	AC0	設定 CG RAM 位址，以開始接收或傳送 CG RAM 的資料
DD RAM 位址設定	0	0	1	AD6	AD5	AD4	AD3	AD2	AD1	AD0	設定 DD RAM 位址，以開始接收或傳送 DD RAM 的資料
忙線旗號 / 位址讀取	0	1	BF	A6	A5	A4	A3	A2	A1	A0	讀取 LCD 內部的 BF 旗號和位址計數器內容
資料寫到 CG RAM 或 DD RAM	1	0	D7	D6	D5	D4	D3	D2	D1	D0	將資料寫到 DD RAM 或 CG RAM
從 CG RAM 或 DD RAM 讀出資料	1	1	D7	D6	D5	D4	D3	D2	D1	D0	從 CG RAM 或 DD RAM 讀取資料
<p>*:Don't care bit</p> <p>I/D=1:遞加 I/D=0:遞減</p> <p>C=1:游標 ON C=0:游標 OFF</p> <p>R/L=1:向右移動 R/L=0:向左移動</p> <p>S=1:顯示器移動 S=0:顯示器不移動</p> <p>B=1:閃爍 B=0:不閃爍</p> <p>DL=1:8 位元 DL=0:4 位元</p> <p>D=1:要顯示 D=0:不顯示</p> <p>S/C=1:顯示器移動 S/C=0:游標要移動</p> <p>N=1:二行 N=0:一行</p> <p>BF=1:LCD 內部正忙線 BF=0:LCD 準備好接受資料</p> <p>F=1:5×10 點矩陣 F=0:5×7 點矩陣</p>											

## 2-2-5 LCD 初始化副程式

LCD 模組在送電後,使用之前,單晶片必須要先去規劃 LCD 的各項功能,或




工作模式,LCD 才能正常工作.這個動作稱為 LCD 的初始化.

例:功能設定命令


圖 2-4-7 LCD 初始化步驟

## 2-3 掃描鍵盤

鍵盤的輸入採用掃描法,由 8051 的埠 3(P3)控制,利用 8051 提供的單一位元定址 I/O,將 P3.0接到鍵盤1, P3.1接到鍵盤2,.....發出掃描信號,而P3.4接到鍵盤A, P3.5接到鍵盤B,.....接收掃描信號.掃描時,由埠 P3.0發出一個低電位,8051依序檢查信號接收端,若接收到低電位,表示按鍵被按下,否則收到高電位.繼續下一個.讀取鍵盤時,遇到的主要問題為彈跳,用軟體去彈跳,也就是若偵測到按鍵被按下,先 delay 18ms.再進行下一個動作.



## 2-4 撥號電路



如電路圖所示，當警報被觸發後，8051 的 P1.3 接到 LS1 繼電器使繼電器線圈有 5V 的電壓差，電話接座的訊號由繼電器 4、3 號腳接通到 ST-71 線圈，再透過 R9 電阻使電話接通，接通後 8051 的 P0.0、P0.1、P1.2 提供串列訊號給撥號 IC HT9200A，撥號表如下：


### HT9200A 撥號表

Table 1: Digits vs. Input Data vs. Tone Output Frequency (Serial Mode)

Digit	D4	D3	D2	D1	D0	Tone Output Frequency (Hz)
1	0	0	0	0	1	697+1209
2	0	0	0	1	0	697+1336
3	0	0	0	1	1	697+1477
4	0	0	1	0	0	770+1209
5	0	0	1	0	1	770+1336
6	0	0	1	1	0	770+1477
7	0	0	1	1	1	852+1209
8	0	1	0	0	0	852+1336
9	0	1	0	0	1	852+1477
0	0	1	0	1	0	941+1336
*	0	1	0	1	1	941+1209
#	0	1	1	0	0	941+1477
A	0	1	1	0	1	697+1633
B	0	1	1	1	0	770+1633
C	0	1	1	1	1	852+1633
D	0	0	0	0	0	941+1633
—	1	0	0	0	0	697
—	1	0	0	0	1	770
—	1	0	0	1	0	852
—	1	0	0	1	1	941
—	1	0	1	0	0	1209
—	1	0	1	0	1	1336
—	1	0	1	1	0	1477
—	1	0	1	1	1	1633
DTMF OFF	1	1	1	1	1	—

Note: The codes not listed in Table 1 are not used D4 is MSB

假設我們現在要撥出 1 的號碼，由上表得知 1 的編碼為 00001，所以利用 HT9200A 所訂定的串列規格給資料，串列時序圖如下：



CLK 是 CLOCK 腳，DATA 是要編碼的資料(共 5 碼) 1，CE 是致能腳

以下是範例程式片段：

TELOUT:

=====片段 1=====

CLR CE ;使 CE 腳為低電位

SETB CLK ;使 CLK 腳為高電位

CALL DLL ;延時

CALL DLL

CALL DLL

CALL DLL

CALL DLL

CALL DLL

=====

=====片段 2=====

MOV A,#00000001B ;撥號數字 1 的編碼

MOV R2,#5 ;共 5 碼

TEL2:

CLR C ;C=0

RRC A ;把 A=00000001 往右轉一位後，C=1，  
;A=00000000

MOV DAT,C ;把 C=1 的資料 MOVE 到 HT9200 的 DATA

腳

CALL DLL ;延時

CLR CLK


CALL DLL

=====

SETB CLK

DJNZ R2,TEL2

RET



我們再舉一個例子來看：由專題的程式碼得知

89C51 的暫存器 40H~49H 存入我們要撥出的 10 位數字

以手機號碼 0921646726 來說，40H=0、41H=9、42H=2、43H=1、44H=6、

45H=4、46H=6、47H=7、48H=2、49H=6，實際撥號程式如下：

;-----

; 電話撥號副程式

;-----

TEL:CLR P3.5

CLR CE

SETB CLK

CALL DELAY4

MOV R0,#40H

MOV R3,#10

TL2: CLR P3.3

MOV 3AH,@R0 ;R0 從 40H 到 49H 十個撥號號碼

CALL TELOUT

SETB P3.3

MOV 3AH,#FFH

CALL TELOUT ;呼叫編碼撥出副程式

INC R0

DJNZ R3,TL2 ;十個撥號號碼，所以十個迴圈，跳至 TL2

SETB CE

CALL DLL

CALL DLL

RET

;=====編碼撥出副程式=====

TELOUT:

```
CLR    CE
SETB   CLK
CALL   DLL
MOV    A,3AH
MOV    R2,#5
```

TEL2:

```
CLR    C
RRC    A
MOV    DAT,C
CALL   DLL
CLR    CLK
CALL   DLL
SETB   CLK
DJNZ   R2,TEL2
RET
```

DLL:

```
MOV    R5,#1
DLL6:  MOV R6,#100
DLL7:  MOV R7,#100
DLL8:  DJNZ R7,DLL8
        DJNZ R6,DLL7
        DJNZ R5,DLL6
RET
```

## 2-5 MQ-2 氣體感測器

### 技術參數

MQ-2具有良好靈敏度和偵測氣體寬廣範圍的特性，其工作點設計在5V加熱電源，而電路電壓不得超過5V。

電壓額定值的規格，範圍很廣，這樣可使成本降低，而可靠性提升。因為他對甲烷、丙烷和天然氣具有高敏感度，所以MQ-2對瓦斯和LPG監視極為實用同時有短而穩定的出始期和高可靠度的特性。

### 特點：

1. 廣泛的探測範圍
2. 高靈敏度/快速響應恢復
3. 優良穩定性/壽命長
4. 簡單驅動電路

### 應用：

可用於家和工廠的氣體泄漏監測裝置，適宜於液化氣、丁烷、丙烷、甲烷、酒精、氫氣、煙霧等的探測，因此它是室內感應最佳的感測器。

### A. 標準工作條件

符號	參數名稱	技術條件	備註
$V_C$	回路電壓	$5V \pm 0.1$	AC OR DC
$V_H$	加熱電壓	$5V \pm 0.1$	AC OR DC
$R_L$	負載電阻	可調	
$R_H$	加熱電阻	$32\Omega \pm 10\%$	室溫
$P_H$	加熱耗能	小於800毫瓦	

**MQ-2**偵測器阻抗的改變，間接的改變將出現在負載電阻 $R_L$ 上，但是當瓦斯氣體如甲烷、丙烷等接觸到偵測器表面時，若有瓦斯濃度的出現，偵測器阻抗立即降低，而由AC(交流)或DC(直流)電源電壓 $V_C$ 和 $V_H$ 在 $R_L$ 上所產生的電壓則是相同的。

### B. 環境條件

符號	參數名稱	技術條件	備註
$T_{ao}$	使用溫度	$-20^{\circ}\text{C} \sim 50^{\circ}\text{C}$	
$T_{as}$	儲存溫度	$-20^{\circ}\text{C} \sim 70^{\circ}\text{C}$	
$R_h$	相對溫度	小於95%Rh	
$O_2$	氧氣濃度	21%(標準條件) 氧氣濃度會影響靈敏度特性	最小值大於2%

### C. 靈敏度特性



符號	參數名稱	技術條件	備註
$R_s$	敏感體表面電阻	$3K\Omega-30K\Omega$ (1000ppm isobutene)	探測濃度範圍 200ppm-5000ppm 液化氣和丙烷
G(3000/10000) isobutane	濃度斜率		300ppm-5000ppm 丁 烷
標準工作條件	溫度： $20^{\circ}C\pm 2^{\circ}C$ 相對溫度： $65\%\pm 5\%$	$V_c : 5V\pm 0.1$ $V_b : 5V\pm 0.1$	5000ppm-2000ppm 甲烷
預熱時間	不小於24小時		300ppm-5000ppm 氫 氣 100ppm-2000ppm 酒 精

#### D. 結構 外形 測試電路

	部件	材料
1	氣體敏感層	二氧化錫
2	電極	金(Au)
3	測量電線引路	鉑(Pt)
4	加熱器	鎳鉻合金(Ni-Cr)
5	陶瓷管	三氧化二鋁
6	防爆管	100 目雙層不銹鋼 (SUB316)
7	卡環	鍍鎳銅材(Ni-Cu)
8	基座	電木
9	針狀管腳	鍍鎳銅材(Ni-Cu)

MQ-2其體積如煙頭般大小，主要組成元件是二氧化錫半導體，這半導體物質和電極組成圓形管狀殼內。

### MQ 系列氣體感測器結構原理圖



MQ-2 氣敏元件的結構和外形如圖一所示(結構A or B),由微型AL2O3 陶瓷管、SnO<sub>2</sub> 敏感層，測量電極和加熱器構成敏感元件固定在塑膠或不銹鋼製成的腔體內，加熱器？氣敏元件提供了必要的工作條件。封裝好的氣

敏元件有6只針狀管腳，其中4個用於信號取出，2個用於提供加熱電流。

### E.靈敏度特性曲線

圖3給出了MQ-2型氣敏元件的靈敏度性。

其中：

溫度：20°C

相對溫度：65%

氧氣濃度：21%

$R_L=5K\Omega$

$R_s$ ：元件在不同氣體，不同濃度下的電阻值。

$R_0$ ：元件在潔淨空氣中的電阻值。




圖4 給出了MQ-2型氣敏元件的溫濕度特性

$R_0$ ：20°C，33%RH 條件下，1000ppm 氫氣中元件電阻。

$R_s$ ：不同溫度，濕度下，1000ppm 氫氣中元件電阻。

## 靈敏度調整

MQ-2 型氣敏元件對不同種類，不同濃度的氣體有不同的電阻值。因此，在使用此類型氣靈敏元件時，靈敏度的調整是很重要的。我們建議您用1000ppm液化氣或1000ppm 丁烷校準感測器。

當精確測量時，報警點的設定應考慮溫濕度的影響。

圖3：





Fig 2 sensitivity characteristics of the MQ-2

## 感測電路


使用電壓比較 IC LM393，暖機完成時(大約 1 分鐘)，調整 R3 可調電阻使 LM393 第 2 腳電壓大於 LM393 第 3 腳的電壓。

一旦瓦斯感測器感測到氣體時，阻值下降，和 3K 電阻分壓後的電壓上升，超過 LM393 第 2 腳電壓時，LM393 的第 1 腳由 LOW 變 HIGH，8051 的 P1.5 為感測腳，由程式判斷是否有感測到瓦斯。



# 第三章 89C51程式規劃

## 3-1 系統方塊圖



## 3-2 動作解說

我們將學習到如何利用8051的輸入、輸出腳來寫程式，並與週邊IC做連結傳資料，進而達到自動偵測、自動撥號的功能。


輸入週邊有：4X4鍵盤、磁簧開關

1. 4X4鍵盤用來設定當產生警報時，撥出的號碼
2. 磁簧開關為位置感應，當門的位置移動時，磁簧開關的訊號產生變化通知8051，8051再作判斷及處理控制撥號
3. 溫度可由鍵盤設定，當超過溫度時，8051控制撥號
4. 當感測到瓦斯時，8051控制撥號

輸出的週邊有：繼電器、撥號IC、LCD

當8051判定有警報時，繼電器導通讓電話接通，電話撥號IC開始產生撥號碼音訊，通知使用者。

### 3-3 程式流程圖



### 3-4 程式碼

```
EN      REG    P1.1      ;LCD 腳位編排
RS      REG    P1.0      ;LCD 腳位編排
SPK     REG    P0.3      ;B聲腳位編排
CLK     REG    P0.1      ;HT9200控制腳
DAT     REG    P0.0      ;HT9200控制腳
CE      REG    P1.2      ;HT9200控制腳
CNT     EQU    38H
        ORG    00H      ;設定程式開頭位置

        MOV    40H,#FFH ;電話號碼初值
        MOV    41H,#FFH ;電話號碼初值
        MOV    42H,#FFH ;電話號碼初值
        MOV    43H,#FFH ;電話號碼初值
        MOV    44H,#FFH ;電話號碼初值
        MOV    45H,#FFH ;電話號碼初值
        MOV    46H,#FFH ;電話號碼初值
        MOV    47H,#FFH ;電話號碼初值
        MOV    48H,#FFH ;電話號碼初值
        MOV    49H,#FFH ;電話號碼初值
```

```

MOV  4AH,#FFH ;電話號碼初值

MOV  50H,#50      ;50H設為50

MOV  51H,#100    ;51H設為100

MOV  5CH,#4      ;設定溫度初值十位數4

MOV  5DH,#0      ;設定溫度初值個位數0

MOV  A,5CH       ;4X10+0=40
MOV  B,#10
MUL  AB
ADD  A,5DH

MOV  5EH,A       ;存到5EH

CLR  20H.1       ;啟動警報功能時 20H.1設為1，不啟動設為0

    AJMP  START

;-----
; MAIN
;-----
START: MOV  SP,#60H
      SETB CE

      ACALL INIT           ;呼叫LCD初始化副程式

WAIT:

      JB  P1.5,WAIT ;瓦斯感測器需暖機 當P1.5為0時，表示暖機完成，程式
      往下，否則跳WAIT

      CALL DELAY1S

L0:

      ACALL ON             ;顯示功能表副程式

L1:

```

```
JNB 20H.1,L1A ;啟動警報功能？ 是則程式往下，否則跳L1A  
JNB P1.4,L1A ;磁簧開關偵測，有則程式往下，無則跳到L1A  
JMP TEL ;跳到撥號副程式
```

L1A:

```
DJNZ 50H,L1B ;50H減1不為0時跳L1B，為0時程式往下  
MOV 50H,#50 ;50H設為50  
DJNZ 51H,L1B ;51H減1不為0時跳L1B，為0時程式往下  
MOV 51H,#100 ;51H設為100  
CALL SHOWTEMP ;呼叫顯示溫度副程式
```

```
JNB 20H.1,L1C ;啟動警報功能？ 是則程式往下，否則跳L1C  
CLR C  
MOV A,5EH ;是否超溫？ 5EH=溫度設定值  
MOV B,5BH ;5BH=目前溫度值  
SUBB A,B  
JNC L1B ;是則程式往下，否則跳L1B  
JMP TEL ;跳到撥號副程式
```

L1B:

```
JNB 20H.1,L1C ;啟動警報功能？ 是則程式往下，否則跳L1C  
JNB P1.5,L1C ;有瓦斯程式往下，否則跳L1C  
JMP TEL ;跳到撥號副程式
```

L1C:

ACALL SCAN ;呼叫掃描鍵盤副程式

MOV A,R5 ;R5用來判斷使用者按何鍵

CJNE A,#BH,L2 ;HELP

AJMP HELP

L2: CJNE A,#CH,L3 ;A鍵

AJMP F1 ;跳到F1

L3: CJNE A,#DH,L4 ;B鍵

AJMP F2 ;跳到F2

L4: CJNE A,#EH,L1 ;C鍵

CPL 20H.1 ;反相一次 20H.1

CALL ALONF ;呼叫ALONF

AJMP L1 ;跳到F1

;-----  
; HELP  
;-----  
HELP:

SJMP L0

ALONF:

JB 20H.1,ALA

MOV A,#8FH ;顯示在LCD第1行第16字

```

        ACALL    W_INS        ;設定LCD顯示位置

MOV    A,#' '
CALL  W_DATA
RET
ALA:
        MOV     A,#8FH        ;顯示在LCD第1行第16字

        ACALL   W_INS        ;設定LCD顯示位置

MOV    A,#'A'
CALL  W_DATA
RET

;-----
;    設定號碼副程式
;-----
F1:
    MOV    R2,#10        ;R2=10
    MOV    R0,#40H        ;存10位電話號碼，存入40H~4AH

ST3:   MOV    @R0,#FFH ;電話初值(無效撥號值)
        INC    R0
        DJNZ  R2,ST3      ;10次跳S3

F3C:
        MOV     A,#80H        ;顯示在LCD第1行第1字

        ACALL   W_INS        ;設定LCD顯示位置

MOV    CNT,#16
MOV    DPTR,#MSG4 ;Input number:
ACALL  W_STR

```

```

MOV      A,#C0H      ;顯示在LCD第2行第1字
ACALL   W_INS       ;設定LCD顯示位置
MOV      CNT,#16
MOV      DPTR,#MSG3 ;顯示空白
ACALL   W_STR
MOV      A,#C2H      ;顯示在LCD第2行第3字
ACALL   W_INS       ;設定LCD顯示位置
MOV      R0,#40H    ;R0指向=40H
F31:    ACALL  SCAN   ;呼叫掃描鍵盤副程式
MOV      A,R5       ;R5為掃描值，用來判斷使用者按何鍵
XRL     A,#16       ;若掃描值等於16不動作
JZ      F31
MOV     A,R5
CJNE   A,#AH,F32    ;按下F鍵，是則往下，否則跳F32
AJMP   SETOK
F32:   CJNE  A,#BH,F33 ;按下E鍵，是則往下，否則跳F33
AJMP   F31
F33:   CJNE  A,#CH,F34 ;按下A鍵，是則往下，否則跳F34
AJMP   F31
F34:   CJNE  A,#DH,F35 ;按下B鍵，是則往下，否則跳F35
AJMP   F31
F35:   CJNE  A,#EH,F36 ;按下C鍵，是則往下，否則跳F36
AJMP   F31

```

```

F36:   CJNE   A,#FH,F37       ;按下D鍵，是則往下，否則跳F37
      AJMP   F31
F37:   MOV    A,R5
      MOV    @R0,A          ;將掃描值存入R0所指向的暫存器位置
      ORL   A,#30H         ;將掃描值轉成ASCII碼
      ACALL  W_DATA        ;傳單字至LCD SHOW出資料
      INC   R0             ;R0指向+1
      CJNE  R0,#4AH,F38    ;R0指向=4AH?不是則跳F38，是則往下
      MOV   R0,#40H        ;R0指向=40H
      MOV   A,#C2H         ;顯示在LCD第2行第3字
      ACALL  W_INS         ;設定LCD顯示位置
F38:   SJMP   F31

SETOK:
      ACALL  CLRLCD
      MOV   A,#84H         ;顯示在LCD第1行第5字
      ACALL  W_INS         ;設定LCD顯示位置
      MOV   CNT,#8
      MOV   DPTR,#MSG9
      ACALL  W_STR
      CALL  DELAY1S
      JMP   L0

;-----
;   設定溫度副程式

```

;-----  
F2:

```
MOV    A,#80H      ;顯示在LCD第1行第1字
ACALL  W_INS       ;設定LCD顯示位置
MOV    CNT,#16
MOV    DPTR,#MSG5 ;Input Tempe:
ACALL  W_STR
MOV    A,#C0H      ;顯示在LCD第2行第1字
ACALL  W_INS       ;設定LCD顯示位置
MOV    CNT,#16
MOV    DPTR,#MSG3 ;顯示空白
ACALL  W_STR
MOV    A,#C2H      ;顯示在LCD第2行第3字
ACALL  W_INS       ;設定LCD顯示位置
MOV    R0,#5CH    ;R0指向=5CH
F21:   ACALL  SCAN   ;呼叫掃描鍵盤副程式
MOV    A,R5        ;R5為掃描值，用來判斷使用者按何鍵
XRL   A,#16        ;若掃描值等於16不動作
JZ    F21
MOV    A,R5
CJNE  A,#AH,F22    ;按下F鍵，是則往下，否則跳F22
AJMP  SETOK2
```

```

F22:  CJNE    A,#BH,F23      ;按下E鍵，是則往下，否則跳F23
      AJMP    F21
F23:  CJNE    A,#CH,F24      ;按下A鍵，是則往下，否則跳F24
      AJMP    F21
F24:  CJNE    A,#DH,F25      ;按下B鍵，是則往下，否則跳F25
      AJMP    F21
F25:  CJNE    A,#EH,F26      ;按下C鍵，是則往下，否則跳F26
      AJMP    F21
F26:  CJNE    A,#FH,F27      ;按下D鍵，是則往下，否則跳F27
      AJMP    F21
F27:  MOV     A,R5
      MOV     @R0,A          ;將掃描值存入R0所指向的暫存器位置
      ORL    A,#30H          ;將掃描值轉成ASCII碼
      ACALL   W_DATA        ;傳單字至LCD SHOW出資料
      INC    R0             ;R0指向+1
      CJNE   R0,#5EH,F28    ;R0指向=5EH?不是則跳F28，是則往下
      MOV    R0,#5CH        ;R0指向=5CH
      MOV    A,#C2H          ;顯示在LCD第2行第3字
      ACALL   W_INS         ;設定LCD顯示位置
F28:  SJMP    F21

```

```

SETOK2:

```

```

    MOV    A,5CH
    MOV    B,#10

```

```

MUL  AB
ADD  A,5DH
MOV  5EH,A      ;存成溫度設定值

```

```

        ACALL  CLRLCD
MOV     A,#84H      ;顯示在LCD第1行第5字
        ACALL  W_INS      ;設定LCD顯示位置
MOV     CNT,#8
MOV     DPTR,#MSG9
        ACALL  W_STR
CALL   DELAY1S
JMP    L0

```

;-----

; 電話撥號副程式 ;詳細請見 HT9200A的介紹

;-----

TEL:

TE3:

```

CALL  ALARM      ;LCD顯示有警報正在撥號
MOV   P0,#11111111B
CLR   P1.3      ;接通電話
CALL  DELAY2
CLR   CE
SETB  CLK
CALL  DELAY2
MOV   R0,#40H   ;電話暫存在40H~4AH裡
MOV   R3,#10    ;共10個數字的電話碼

```

TL2:

```

MOV   A,@R0

```

```

    CJNE  A,#0,TLL      ;號碼有0時，變成A，見HT9200資料
    MOV   A,#0AH
TLL:
    MOV   3AH,A
    CALL  TELOUT        ;控制撥號IC HT9200撥號

    MOV   3AH,#FFH     ;分格訊號
    CALL  TELOUT
    INC   R0
    DJNZ  R3,TL2       ;跳到TL2十次

    SETB  CE           ;結束撥號

    CALL  DLL
    CALL  DLL

    CALL  DELAY1S      ;延時

    CALL  DELAY1S
    CALL  DELAY1S

    CALL  ARMON        ;撥放警告音

    SETB  P1.3         ;掛掉電話

    CALL  DELAY1S
    CALL  DELAY1S
    JMP   L0

TELOUT:                ;見HT9200資料

    CLR   CE
    SETB  CLK
    CALL  DLL
    CALL  DLL

```

```

CALL DLL
CALL DLL
CALL DLL
CALL DLL
MOV A,3AH
MOV R2,#5
TEL2:
CLR C
RRC A
MOV DAT,C
CALL DLL
CLR CLK
CALL DLL
SETB CLK
DJNZ R2,TEL2
RET

```

```

DLL:
MOV R5,#1
DLL6:MOV R6,#50
DLL7:MOV R7,#50
DLL8:DJNZ R7,DLL8
DJNZ R6,DLL7
DJNZ R5,DLL6
RET

```

SHOWTEMP:

ACALL ATOD ;呼叫副程式，致能ADC0804這顆IC，並把數值讀進

來P2腳

MOV 52H,B

WAIT1:MOV R3,#00H ;令字形及溫度值的指標為0

```

LOOP1:CLR C          ;設C為0

        ACALL READTEMP      ;讀溫度資料表，查表法

        INC R3              ;把字形及溫度值的指標加1

        MOV A,52H          ;把量測到的溫度數值放進去A

        SUBB A,R1          ;R1是18度到40度查表法查到的數值，A是量測到的A/D
數值，A減去R1

        JNC LOOP1          ;A減去R1(從18度到40度)，一直減到 R1(查表)
比A(量測)大 <=====C被借位變成1

        ;此時溫度已找到

        DEC R3              ;多加了兩次要減回來

        DEC R3

        ACALL DISPLAY      ;呼叫顯示副程式

        RET
; =====
; == 顯示副程式 ==
; =====
DISPLAY:
        MOV A,#C6H        ;設定顯示位置為第2行第7個字

        ACALL W_INS        ;送出指令到LCD

        CLR C

        MOV A,R3          ;R3加20

        ADD A,#20

```

```

MOV  5BH,A    ;
MOV  B,#10
DIV  AB
ADD  A,#30H   ;轉成ASCII碼

ACALL W_DATA   ;送出資料到LCD

MOV  A,B     ;取出個位數

ADD  A,#30H   ;轉成ASCII碼

ACALL W_DATA   ;送出資料到LCD

RET
; =====
; ==  A/D轉換副程式 ==
; =====
; 起動ADC0804
ATOD:
CLR  P1.7     ;令ADC0804的WR=0

SETB P1.7     ;令ADC0804的WR=1

;等待ADC0804把類比電壓轉換成數位資料
WAITT:
JB  P1.6,WAITT ;等待ADC0804的INTR=0

;讀進轉換完成的Vx對應值

MOV  A,P2     ;把轉換完成的數位資料讀進累積器A內

MOV  B,A     ;返品主程式

RET

```

```

; =====
; =讀溫度資料副程式 =
; =====

```

READTEMP:MOV DPTR,#TTABLE;取出TTABLE 溫度值的表

```
MOV A,R3 ;查表位置->A
```

```
MOVC A,@A+DPTR ;依位置取出->A
```

```
MOV R1,A ;A->R1
```

```
RET
```

```

; =====
; == 溫度資料表 ==
; =====

```

TTABLE:

```

DB 110
DB 112
DB 115
DB 118
DB 120
DB 123
DB 125
DB 128
DB 130
DB 133
DB 135
DB 138
DB 140
DB 142
DB 145
DB 148
DB 150
DB 152

```

DB 155  
DB 157  
DB 159  
DB 161  
DB 163  
DB 165  
DB 167  
DB 169  
DB 171  
DB 173  
DB 175  
DB 177  
DB 179  
DB 181  
DB 183  
DB 185  
DB 187  
DB 189  
DB 191  
DB 193  
DB 195  
DB 197  
DB 199  
DB 201  
DB 203  
DB 205  
DB 207  
DB 209  
DB 211  
DB 213  
DB 215  
DB 217  
DB 219

;----- --  
; SCAN KEY  
;----- --

```

SCAN:  MOV    R4,#00H           ;清除R4，R4存放掃描按鍵值

        MOV    R7,#4           ;R7設定為4，掃描4次P3.0~P3.3

        MOV    R6,#1111110B    ;從P3.0開始掃描

COL:    MOV    A,R6
        MOV    P3,A

        JB     P3.4,S1         ;無按鍵(1)~S1，有按鍵(0)往下執行

        JNB    P3.4,$          ;消除彈跳誤動作，即按鍵(0)等待放開

```

按鍵(1)往下執行

```

        ACALL  DELAY
        ACALL  DELAY
        ACALL  DELAY
        SJMP   S5      ;First Line
S1:     INC    R4
        JB     P3.5,S2
        JNB    P3.5,$      ;DEBOUNCE
        ACALL  DELAY
        ACALL  DELAY
        ACALL  DELAY
        SJMP   S5      ;Second Line
S2:     INC    R4
        JB     P3.6,S3
        JNB    P3.6,$
        ACALL  DELAY
        ACALL  DELAY
        ACALL  DELAY
        SJMP   S5      ;Third Line
S3:     INC    R4
        JB     P3.7,S4
        JNB    P3.7,$
        ACALL  DELAY
        ACALL  DELAY

```

```

ACALL    DELAY
        SJMP    S5    ;Fouth Line
S4:     INC     R4
        MOV     A,R6

        RL     A      ;換P3.1列開始掃描

        MOV     R6,A
        DJNZ   R7,COL

S5:     CJNE   R4,#00H,S6
        MOV    R5,#01H
        RET
S6:     CJNE   R4,#01H,S7
        MOV    R5,#04H
        RET
S7:     CJNE   R4,#02H,S8
        MOV    R5,#07H
        RET
S8:     CJNE   R4,#03H,S9
        MOV    R5,#0AH
        RET
S9:     CJNE   R4,#04H,SA
        MOV    R5,#02H
        RET
SA:     CJNE   R4,#05H,SB
        MOV    R5,#05H
        RET
SB:     CJNE   R4,#06H,SC
        MOV    R5,#08H
        RET
SC:     CJNE   R4,#07H,SD
        MOV    R5,#00H
        RET
SD:     CJNE   R4,#08H,SE
        MOV    R5,#03H
        RET
SE:     CJNE   R4,#09H,SF

```

```

MOV R5,#06H
RET
SF: CJNE R4,#0AH,SG
MOV R5,#09H
RET
SG: CJNE R4,#0BH,SOK
MOV R5,#0BH
RET

```

```

SOK: MOV A,R4
MOV R5,A
RET

```

```

;-----
; POWER ON

```

```

;-----
ON:  PUSH  A

      MOV  A,#80H      ;顯示在LCD第1行第1字

      ACALL W_INS
      MOV  CNT,#16
      MOV  DPTR,#MSG1 ;A:SET NUM.
      ACALL W_STR

      MOV  A,#C0H      ;顯示在LCD第2行第1字

      ACALL W_INS
      MOV  CNT,#16
      MOV  DPTR,#MSG2 ;Tempe:  C /  C
      ACALL W_STR

```

```

MOV  A,#C8H      ;設定顯示位置為第2行第9個字

```

```

ACALL W_INS      ;送出指令到LCD

```

```

MOV  A,#11011111B ;顯示度

```

ACALL W\_DATA ;送出資料到LCD

MOV A,#CCH ;設定顯示位置為第2行第12個字

ACALL W\_INS ;送出指令到LCD

MOV A,5CH ;

ADD A,#30H

CALL W\_DATA

MOV A,5DH ;

ADD A,#30H

CALL W\_DATA

MOV A,#11011111B ;顯示度

ACALL W\_DATA ;送出資料到LCD

CALL ALONF

POP A

RET

ALARM: PUSH A

MOV A,#80H ;顯示在LCD第1行第1字

ACALL W\_INS

MOV CNT,#16

MOV DPTR,#MSGP ;ALARM! CALLING..

ACALL W\_STR

MOV A,#C0H ;顯示在LCD第2行第1字

ACALL W\_INS

MOV R2,#10

MOV R0,#40H

AL2:

```
MOV  A,@R0
ADD  A,#30H
CALL W_DATA
INC  R0
DJNZ R2,AL2
```

```
POP  A
```

```
RET
```

```
;-----
```

```
; INIT RAM & LCD
```

```
;-----
```

```
INIT:  MOV      R0,#6
        MOV      R1,#30H
        CLR      A
NEXT0:  MOV      @R1,A
        INC      R1
        DJNZ     R0,NEXT0
        MOV      R0,#10
        MOV      R1,#40H
        MOV      A,#FFH
NEXT1:  MOV      @R1,A
        INC      R1
        DJNZ     R0,NEXT1
;
        MOV      A,#38H
        ACALL    W_INS
        MOV      A,#38H
        ACALL    W_INS
        MOV      A,#38H
        ACALL    W_INS
```

```

MOV     A,#38H
ACALL  W_INS
MOV     A,#08H ;FUNCTION SET
ACALL  W_INS
MOV     A,#01H ;CLEAR DISPLAY
ACALL  W_INS
MOV     A,#06H ;INPUT MODE SET
ACALL  W_INS
MOV     A,#0CH ;DISPLAY ON/OFF
ACALL  W_INS
RET

;-----
; CLEAR LCD
;-----
CLR_LCD: MOV     A,#01H
          ACALL  W_INS
          RET

;-----
; WRITE STRING TO LCD
;-----
W_STR:   PUSH    A
          MOV     R5,#0
STR:     MOV     A,R5
          MOVC   A,@A+DPTR
          ACALL  W_DATA
          INC    R5
          DJNZ  CNT,STR
          POP    A
          RET

;-----
; WRITE INS TO LCD
;-----
W_INS:  PUSH    A
          ACALL  CHECK
          CLR    RS
          MOV    P0,A

```

```

        SETB    EN
        CLR     EN
        POP     A
        RET

;-----
; WRITE DATA TO LCD
;-----
W_DATA: PUSH    A
        ACALL   CHECK
        SETB    RS
        MOV     P0,A
        SETB    EN
        CLR     EN
        POP     A
        RET

;-----
; CHECK LC D BUSY
;-----
CHECK:  PUSH    A
BUSY:
        ACALL   DELAY
        POP     A
        RET

;-----
; DELAY
;-----
DELAY:  MOV     R6,#20
D1:     MOV     R7,#150
        DJNZ   R7,$
        DJNZ   R6,D1
        RET

; =====
; ==延時副程式R5來定時==
; =====

```

```
DELAY2: MOV R5,#5
DL0: MOV R6,#200
DL1: MOV R7,#200
DL2: DJNZ R7,DL2
      DJNZ R6,DL1
      DJNZ R5,DL0
      RET
```

```
DELAY1S: MOV R5,#20
DDL0: MOV R6,#200
DDL1: MOV R7,#200
DDL2: DJNZ R7,DDL2
      DJNZ R6,DDL1
      DJNZ R5,DDL0
      RET
```

```
ARMON:
      MOV R2,#50
ARL:
      CALL BZ
      CALL DELAY2
      DJNZ R2,ARL
      RET
```

BZ:

```
      MOV R6,#0
B2:   CALL DE1
      CPL SPK
      DJNZ R6,B2
      SETB SPK
```

```
      RET
```

```
;-----
```

```


DE1:   MOV R7,#100
$1:    NOP
       DJNZ R7,$1
       RET

;-----
; MESSAGE
;-----
MSG1:  DB      'A:SET NUM.      '
MSG2:  DB      'Tempe:  C /  C'
MSG3:  DB      '                '
MSG4:  DB      'Input number:  '
MSG5:  DB      'Input Tempe:   '
MSG9:  DB      'SUCCESS!'
MSGC:  DB      'CALLING...!'
MSGP:  DB 'ALARM! CALLING..!'
       END


```

# 3-5 電路圖


## 8051 主電路




## 磁簧開關電路




## 5V 穩壓電路




## LCD 電路




## 溫度感測電路




## 瓦斯感測電路





## 電話撥號電路



## 鍵盤電路





3-

### 3-6 零件表

項目	數量	編號	規格	備註
1	1	1	煙霧感測器	MQ-2
2	1	C1	220uF/16V	電解電容
3	2	C2, C4	104	陶瓷電容
4	1	C3	150P	陶瓷電容
5	4	C5, C6, C9, C10	30PF	陶瓷電容
6	3	C7, C8, C11	10uF	電解電容
7	1	D1	LED	發光二極體
8	1	J1	熱敏電阻	
9	1	J2	POWER	變壓器插座
10	1	J3	磁簧開關	
11	1	J4	電話接座	
12	2	LS1, L1	ST-71	線圈
13	1	RN1	10K 排阻	
14	2	R1, R4	1K	電阻
15	2	R7, R2	10K	電阻

16	1	R3	50K	電阻
17	1	R5	3K	電阻
18	1	R6	3.9K	電阻
19	2	R9, R8	470	電阻
20	1	U1	LCM	LCD 顯示器
21	1	U2	78M05	穩壓 IC
22	1	U3	ADC0804	AC 轉 DC IC
23	1	U4	8051	單晶片
24	1	U5	KEYING	4*4 鍵盤
25	1	U6	LM393	電壓比較 IC
26	1	U7	HT9200A	電話撥號 IC
27	1	Y1	12MHZ	震盪器
28	1	Y2	3.579MHZ	震盪器

# 附錄

## 作者簡介

周晉玄

這次專題一開始老師給了我們很多建議，就算是舊的題目能夠加上自己的創意，那就是自己的作品，於是我們就朝著這方向前進，想要結合灑水系統或自動化家電等功能。但到之後因為我要準備公職考試，時間及金錢因素的考量下，最後折衷做了此款。

在此要感謝我們的指導老師趙維和老師，能為我們指點讓我們目標明確，也要謝謝我的組員，能盡量配合我的時間並提醒專題的進度，在分工上幾乎可以說是互補的，當成品完成時，那份成就感不是三言兩語可以形容的。

林泉助

起初，對專題都有不少想法，想要功能越多越好，但實際要著手進行時，發現我們把目標定得太大，要在有限的時間內完成的可能相對減小，所以經討論後選擇幾個最重要的功能。製作過程也不是非常順利，尤其是它完全沒反應又找不出原因時，真的讓人有點不知所措，不過經過不斷的修正及測試，終於完成了。在成功的那一刻，有股莫名的感動湧上心頭，我想，那就是所謂的成就感吧。

## 參 考 文 獻

- [1] 吳朗，感測與轉換原理元件與應用，全欣資訊圖書
- [2] 蔡朝洋，單晶片微電腦8051原理與應用，全華
- [3] 陳明燦，單晶片8051實作入門，文魁
- [4] 89C51的DATA SHEET